<commit_before>@echo off

set JAVACLASSPATH=%CLASSPATH%;%IBIS_ROOT%\build;

rem From Ant:
rem Slurp the command line arguments. This loop allows for an unlimited number
rem of arguments (up to the command line limit, anyway)
set IBIS_APP_ARGS=%1
if ""%1""=="""" goto doneStart
shift
:setupArgs
if ""%1""=="""" goto doneStart
set IBIS_APP_ARGS=%IBIS_APP_ARGS% %1
shift
goto setupArgs
rem This label provides a place for the argument list loop to break out
rem and for NT handling to skip to.

:doneStart

"%JAVA_ROOT%\bin\java" -classpath "%JAVACLASSPATH%" ibis.impl.nameServer.tcp.NameServer %IBIS_APP_ARGS%<commit_msg>Updated the windows nameserver script<commit_after>@echo off

if "%OS%"=="Windows_NT" @setlocal

set JAVACLASSPATH=%CLASSPATH%;%IBIS_ROOT%\build;

set ConnectHub="-Dibis.connect.enable -Dibis.connect.control_links=RoutedMessages -Dibis.connect.data_links=PlainTCP"

set NS_ARGS=
set Ddebug=
set Dhub=
set Dport=
set Dpoolport=
set Dhubport=

:setupArgs

if ""%1""=="""" goto doneArgs

if "%1"=="-debug" (
    set Ddebug="-Dibis.connect.debug=true -Dibis.connect.verbose=true"
    goto nextarg
)

if "%1"=="-d" (
    set Ddebug=%Ddebug% "-Dibis.name_server.debug=true"
    goto nextarg
)

if "%1"=="-v" (
    set Ddebug=%Ddebug% "-Dibis.name_server.verbose=true"
    goto nextarg
)

if "%1"=="-controlhub" (
    set Dhub=%ConnectHub%
    set NS_ARGS=%NS_ARGS% -controlhub
    goto nextarg
)

if "%1"=="-port" (
    set Dport="-Dibis.name_server.port=%2"
    shift
    goto nextarg
)

if "%1"=="-poolport" (
    set Dpoolport="-Dibis.pool.server.port=%2"
    shift
    goto nextarg
)

if "%1"=="-hubport" (
    set Dhubport="-Dibis.connect.port=%2"
    set Dhub=%ConnectHub%
    shift
    goto nextarg
)

set NS_ARGS=%NS_ARGS% "%1"

:nextarg
    shift
    goto setupArgs

rem This label provides a place for the argument list loop to break out
rem and for NT handling to skip to.

:doneArgs

"%JAVA_ROOT%\bin\java" -classpath "%JAVACLASSPATH%" %Ddebug% %Dhub% %Dport% %Dpoolport% %Dhubport% ibis.impl.nameServer.tcp.NameServer %NS_ARGS%

if "%OS%"=="Windows_NT" @endlocal
{'fraction_non_alphanumeric': 0.14224137931034483, 'fraction_numerical': 0.007662835249042145, 'mean_word_length': 4.468586387434555, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '7297340', 'n_tokens_mistral': 802, 'n_tokens_neox': 787, 'n_words': 203}